

## 第1章 関数型言語 Haskell とは

Haskell は \_\_\_\_\_ と呼ばれ、ラムダ計算を基本としながらも実際の使用に便利な機能を追加したプログラミング言語である。一方 Objective Caml ( 以下では OCaml ) や Standard ML などの ML 系の言語も関数型言語である。

以下のような特徴は ML 系の言語と共通している。

1. \_\_\_\_\_され、実行時の型エラーを完全に防ぐことができる。
  2. \_\_\_\_\_を持ち、プログラマがメモリの管理に煩わされることがない。
  3. 関数を他の関数の引数としたり、関数を生成して他の関数の戻り値としたり、関数を値として使うことができる( \_\_\_\_\_ )。
  4. \_\_\_\_\_により、汎用性のある関数を定義することができる。
  5. \_\_\_\_\_により、(ほとんどの場合) プログラム中に型を書く必要はない。
  6. 再帰的なデータ構造を表すのに便利な \_\_\_\_\_ を定義することができる。
  7. \_\_\_\_\_ を使用して、関数を簡潔に定義することができる。

また、以下のような点が ML 系と異なる。

1. 式に \_\_\_\_\_ はない。入出力や参照の書換えなどの効果も式の値として扱われ、型によって区別される。このため、プログラムの同値性などの性質に関する考察が、より容易になる。
  2. \_\_\_\_\_ を採用し、グラフ簡約によって実行される。
  3. \_\_\_\_\_ など、型システムが（ML系の言語とは異なる方向に）発展している。

## 1.1 なぜ、Haskellを学習するのか？

何よりも、複数のプログラミング言語を学習することは、プログラミングに関する視野を広め、もともと知っていた言語でのプログラミング能力の向上にも役立つ。（複数といっても例えば、CとJavaでは性格が似すぎていてる。）ある言語でひじょうに簡単にできることが、他の言語ではとてつもなく難しいことがある。難しいことが、本質的に難しいのか、非力あるいは不適切な道具のせいで難しいのかを判断できることは重要である。

Haskellは汎用のプログラミング言語であり、どのようなプログラムも原理的には記述することができる。しかし、やはり、得意・不得意はある。Cが得意とする、OSやデバイスドライバなどのようなハードウェアに密着したプログラムや効率が要求される数値計算のプログラム、Javaのようなオブジェクト指向言語が得意とする、ユーザインターフェースのプログラムは Haskell は得意ではない。（もちろん Haskell は進化途中的言語なので、これらの分野に Haskell を応用できるように拡張しようとする研究も活発に（？）進められている。）

Haskellの（MLも）得意な分野は記号処理と呼ばれる分野である。例えば、パズルを解いたり、ゲーム（といっても、シューティングゲームではなく、チェスや将棋・囲碁のようなゲームである）の戦略を考えたりするのも、記号処理である。このような分野のプログラムは Haskell の得意とするところである。

コンパイラやインタプリタなどプログラミング言語の処理系（関数型言語自体の処理系だけではなく、命令型・オブジェクト指向型・論理型など他のパラダイムの言語の処理系も含む）の記述も Haskell は得意としている。

プログラミング言語の意味を数学的に記述する、表示的意味論という手法がある。表示的意味論の目的は、（命令型の）プログラムの意味を数学的概念を用いて記述し、同値性などの性質を厳密に考察できることである。もともとこの分野の研究で、命令型プログラムの意味を関数として記述することが行なわれていた。その成果を素直に Haskell に翻訳すればインタプリタやコンパイラが得られることになる。Haskell によって、命令型言語やオブジェクト指向型言語・論理型言語のインタプリタやコンパイラを作成することにより、副次的な効果としてこれらのプログラミング言語のより深い理解が期待できる。また、\_\_\_\_\_という概念を使いこなすことにより、普通の命令型言語にない変わった副作用も、容易に実装することができる。このため先進的なプログラミング言語のプロトタイプ実装にも適している。

遅延評価は、概念的に \_\_\_\_\_ を扱うことを可能にし、強力なプログラムの部品化の方法を提供する。

Haskellを含む関数型言語は言語の構成要素が（基本的には）単純であるため、型クラスなど型システムの研究が進んでいる。このような先進的な型システムの概念は後で（場合によって少し形を変えて）メインストリームの言語に採用されることがある。

Haskellに関する参考文献・リンクを章末にまとめて紹介する。日本語の書籍もいくつか手に入るようになってきている [11, 12]。

---

---

---

---

## 1.2 Haskell 処理形の入手とインストール

Haskell の処理形としてここでは GHC (Glasgow Haskell Compiler<sup>1</sup>) を利用することにする。GHC は Haskell の処理系の事実上の標準(デファクトスタンダード)である。GHC は、<http://www.haskell.org/ghc/> からダウンロードすることができる。Windows の場合、インストールは入手したファイル (ghc-*version*.msi) をダブルクリックするだけである。

また Linux 用の RPM ファイルなども上記のホームページから入手することができる。

## 1.3 GHCi のコマンド

GHC は、多くのプログラムの集合体であり、gcc や javac のように実行可能形式を出力するバッヂコンパイラ (ghc) もその中に含まれている。

ここでは、その中で対話型の処理系である GHCi(ghci) の使用法を説明する。GHCi を起動すると、

```
Prelude>
```

のようなプロンプトが表示される。このプロンプトからコマンドを入力することによってファイルからプログラムをロードし、式を評価することができる。

GHCi のコマンドには次のようなものがある。

コマンド	省略形	意味
:load <i>file</i>	:l	<i>file</i> をロードする。
:also <i>file</i>	:a	<i>file</i> を追加ロードする。
:reload	:r	以前にロードしたファイルをリロードする。
:type <i>expr</i>	:t	<i>expr</i> の型を表示する。
:cd <i>directory</i>	:c	ディレクトリを変更する。
:! <i>command</i>		<i>command</i> を実行する。
:?		ヘルプを表示する。
:quit	:q	GHCi を終了する。

また、単に式を入力すると、その式を評価してその結果を出力する。

```
Prelude> 1+2
3
```

3 のように斜体になっているところがシステムの出力で、それ以外がユーザの入力である。

Haskell のプログラムファイルには通常 `____` という拡張子をつける。

## 1.4 Haskell のプログラムの基本

Haskell の仕様書は <http://www.haskell.org/> から入手することができる。以下では、Haskell の仕様の基本的なところを紹介する。

<sup>1</sup>Guarded Horn Clauses という論理プログラミング言語と同じ頭文字だが、何の関係もない。

変数の宣言 Haskell では他のプログラミング言語同様、変数を宣言して良く使う式に名前をつけることができる。ただし、C 言語のような命令型言語と異なり、変数は一度宣言するとその値を変えることはできない（代入はできない）。

変数の宣言は次の形で行なう。

変数名 = 式

複数の変数をまとめて宣言する時は次の形式になる。

```
{  
  変数名1 = 式1;  
  変数名2 = 式2;  
  ...;  
  変数名n = 式n  
}
```

つまり、前後を“{”と“}”（プレース）で囲み、“;”（セミコロン）で区切る。ただし、プレースとセミコロンは多くの場合省略でき、以下の例でも省略する場合がある。省略できる条件は付録で説明する。

変数名には C 言語と同じようにアルファベット、数字、“\_”（アンダースコア）が使えるほか、“”（アポストロフィ）も使うことができる。ただし、通常の変数名は \_\_\_\_\_（またはアンダースコア）から始まる必要がある。大文字から始まる名前は、後で紹介する構成子名に用いる。

プログラム Haskell のプログラムはモジュールの集合で、1 つのモジュールは基本的には複数の変数の宣言の前に

```
module モジュール名 where
```

というヘッダ部分をつけた形式である。つまり、以下のようなかたちである。

```
module モジュール名 where {  
  変数名1 = 式1;  
  変数名2 = 式2;  
  ...;  
  変数名n = 式n  
}
```

ただし変数の宣言の他に、import 宣言や型の宣言、型クラス関係の宣言などを書くことができる。これらについては後述する。通常、1 つのファイルに 1 つのモジュールを記述する。

「module モジュール名 where」の部分を省略すると Main という名前のモジュールの定義と解釈される。プレースやセミコロンも多くの場合省略されるので、もっとも簡単な場合、Haskell のプログラムは単に次のような形式をしていることになる。

```
変数名1 = 式1  
変数名2 = 式2  
...  
変数名n = 式n
```

---

---

---

ラムダ式 Haskell では、関数を表すのにラムダ ( $\lambda$ ) 式と呼ばれる記法を用いることができる。ラムダ式はもともとは、ラムダ計算という数学の体系で使われている、名前のない関数を表現するための記法である。例えば、 $\lambda x.x$  は \_\_\_\_\_、 $\lambda xy.y$  は  $x, y$  の 2 引数を受け取って、 $y$  を返す関数を表す。つまり、“ $\lambda$ ” と “.” の間に仮引数を書き、“.” の後に戻り値の式を書く。

ただし、Haskell では、“ $\lambda$ ” の代わりに “\_” ( バックスラッシュ、ただし日本語環境では円記号 (“¥”) が表示されることが多い ) “.” ( ピリオド ) の代わりに “\_\_” を用いる。

ラムダ式	Haskell	OCaml
$\lambda x.x$	_____	fun x -> x
$\lambda xy.y$	_____	fun x y -> y

関数の適用は、関数と実引数を並べて書くだけでよい。通常の数学の記法や C 言語などのように引数に括弧をつける必要はない。

```
c0 = \ f x -> x;
c1 = \ f x -> f x;
c2 = \ f x -> f (f x);
```

実は、このようにラムダ式を名前をつける時は、仮引数を “=” の左辺に並べて、

```
c0 f x = x;
c1 f x = f x;
c2 f x = f (f x);
```

のように書くことができる。(むしろ、このように書く方が普通である。)

コメント Haskell のコメントには 2 つの形がある。 \_\_\_\_\_ という形と \_\_\_\_\_ という形である。

## 1.5 組み込みのデータ型と演算子

Haskell では真偽値 (Bool) 整数 (Int, Integer—Int は固定 (有限) 精度, Integer は無限精度) 浮動小数点数 (Float, Double)、文字 (Char) などは組み込みのデータ型として用意されている。Bool 型のリテラルは \_\_\_\_\_ と \_\_\_\_\_ であり、Integer, Float, Double, Char 型などのリテラルの記法は C 言語とほぼ同様である。

これらのデータ型に対しては組み込みの関数や演算子がいくつか用意されている。Lisp の場合と異なり、算術演算子の “+”, “-”, “\*” などは、 $1+2*3$  のように通常の中置記法で用いる。

if ~ then ~ else 式も組み込みで用意されている。次の形で用いる。

if 式<sub>1</sub> then 式<sub>2</sub> else 式<sub>3</sub>

式<sub>1</sub> は \_\_\_\_\_ でなければならず、式<sub>2</sub> と式<sub>3</sub> は同じ型でなければならない。式<sub>1</sub> が True の時は式<sub>2</sub>、False の時は式<sub>3</sub> として評価される。なお、if ~ then ~ else 式は、Haskell では後で説明するように特殊な評価順を必要とする特殊形式ではない。

次の例は階乗 (factorial) の関数の Haskell での宣言である。

```
fact n = if n==0 then 1 else n*fact(n-1);
```

関数適用は他のどんな中置記法よりも \_\_\_\_\_。そのため、fact(n-1) は fact n-1 と

書くことはできない。後者は \_\_\_\_\_ の意味になってしまう。逆に  $n^*(\text{fact}(n-1))$  の外側の括弧は必要ない。

なお、この例のように変数は \_\_\_\_\_ することが可能である。再帰的定義に特別な文法を使用する必要はない。

問 1.5.1 `fact 100` を計算してみよ。

---

---

リスト リスト型も組み込みのデータ型として用意されている。Scheme の `cons` に対応するのは中置記法の “\_” であり、`nil` に対応するのは \_\_\_\_\_ である。“:” は \_\_\_\_\_ である。つまり、`1:2:[]` は `1:(2:[])` のことを表す。

Scheme	Haskell	OCaml
<code>nil</code>	<code>[]</code>	<code>[]</code>
<code>(cons 1 nil)</code>	<code>1:[]</code>	<code>1::[]</code>
<code>(cons 1 (cons 2 nil))</code>	<code>1:2:[]</code>	<code>1::2::[]</code>

リストのリテラルの記法として、要素を “,” (コンマ) で区切って並べ、“[” と “]” で囲む記法も用意されている。例えば `[1,2,3,4]` は、 \_\_\_\_\_ のことである。

---

リスト型はパラメトリックな型である。つまり、リストの要素の型をパラメータとする。要素の型が `Integer` 型の場合、そのリストの型は \_\_\_\_\_ 型、要素の型が `Double` 型の場合リストの型は \_\_\_\_\_ 型と書き表される。Haskell の文字列 (`String`) 型は実は文字のリスト型として表されている。つまり、

`type String = [Char]`

のように定義されている。ここで、

`type` 型名 = 型

は型の別名 (type alias) を宣言する形式である。上の例の場合 `String` という型名が、`[Char]` という型の別名となる。型名は大文字から始まる識別子でなければならない。

組 組 (tuple) も組み込みのデータ型として用意されている。組は要素を “,” (コンマ) で区切って並べ、“(” と “)” で囲んで表す。組はリストの場合と異なり、要素の型が同一である必要はない。`(1, 'a')` という式の型は \_\_\_\_\_ と表記される。また、`(2, 'b', [3])` という式の型は \_\_\_\_\_ と表記される<sup>2</sup>。

---

<sup>2</sup> 実際の Haskell では、型クラス (type class) というものが関係するため、これらの式はもう少し複雑な型を持つ。ここでは型クラスの説明は避けて、実際よりも単純化した型を紹介する。

関数型 関数の型は “->” という記号を使って、`Integer -> Char` のように表記される。これは、引数の型が `Integer` で戻り値の型が `Char` の関数の型である。“->” は \_\_\_\_\_ である。つまり、`Bool -> Bool -> Bool` という型は \_\_\_\_\_ と解釈される。

これは、実は Haskell の多引数の関数は、“関数を返す関数” として表現されている（このことを \_\_\_\_\_ という）ことを示している。たとえば、`x` と `y` の 2 引数を受け取って、`x+y` を返す関数は、`x` を受け取って “`y` を受け取って、`x+y` を返す関数” を返す関数と同一視されている。

多相型 `Integer` や `Char` のように型定数の名前は必ず大文字から始まる。一方、`a` や `b` のように小文字で始まる識別子が型の中に現れる場合、これらは \_\_\_\_\_ である。これらの型変数は使用する時に、より具体的な型に置き換えることができる。例えば、`[a] -> [b] -> [(a, b)]` という型を持つ関数は、`[Char] -> [Integer] -> [(Char, Integer)]` という型を持つ関数として使用しても良いし、`[String] -> [Integer -> Integer] -> [(String, Integer -> Integer)]` として使用しても良い。

このように型変数を含む型を \_\_\_\_\_ という。Haskell は \_\_\_\_\_ とともに多相型を許す代表的なプログラミング言語である。

なお、変数の型をプログラム中に明示したい場合 “`::`” という記号を使って、

変数名 `::` 型

と書く。例えば `fact` の型を明示しておきたい場合は、

```
fact :: Integer -> Integer;
fact n = if n==0 then 1 else n*fact(n-1);
```

のようにする。ただし通常は、変数の型はプログラマが明示しなくても、Haskell 処理系が推論してくれる。この仕組みを \_\_\_\_\_ という。

## 1.6 パターンマッチング

Haskell では、関数の仮引数の部分に \_\_\_\_\_ というものを書いて、引数の形に応じて場合分けを行なう事が可能である<sup>3</sup>。パターンとは、大雑把に言って変数と定数、構成子からのみ生成される式である。

```
myLength :: [a] -> Integer;
myLength []      = 0;
myLength (x:xs) = 1 + myLength xs;
```

この例の場合、`myLength` の引数が空リスト（`[]`）ならば 1 行目が選択される。一方、1 つ以上の要素を持つリストならば 2 行目が選択され、リストの先頭要素が変数 `x` に束縛され、残りのリストが変数 `xs` に束縛される。（なお、`myLength` とほとんど同じ関数 `length :: [a] -> Int` が標準ライブラリに用意されている。）

パターンマッチングで、右辺で使わない部分には “`_`”（アンダースコア）を使って変数に束縛せず、無視することができる。例えば `myLength` の場合、`x` は右辺で使用していないので、

`myLength (_:xs) = 1 + myLength xs`

<sup>3</sup>一般に関数型言語はデータの種類は増えずに処理（関数）が増えるような場合が得意、オブジェクト指向言語は、データ（クラス）の種類が増えて、処理（メソッド）が増えないような場合が得意である。

と書くことができる。

case ~ of 式もパターンマッチングを行なう。case ~ of 式は次の形で用いる。

```
case 式0 of {  
  パターン1 -> 式1;  
  パターン2 -> 式2;  
  ⋮  
  パターンn -> 式n  
}
```

式<sub>0</sub>を評価して、上から順にパターンと照合し、パターン<sub>1</sub>にマッチするならば式<sub>1</sub>が、パターン<sub>m</sub>にマッチするならば式<sub>m</sub>がそれぞれ評価される。

if 式<sub>1</sub> then 式<sub>2</sub> else 式<sub>3</sub> という式も次の case ~ of 式の略記法と解釈することが可能である。

```
case 式1 of { True -> 式2; False -> 式3 }
```

この他に、let式(後述)やλ式など変数を束縛するところでもパターンを書くこともできる。例えば、

```
\ (x:xs) -> x
```

などである。(この関数はheadという名前で標準ライブラリに用意されている。)この場合は、パターンは一種類のみで場合分けはできず、パターンにマッチしない引数が与えられればエラーとなる。

問 1.6.1 リスト中の数の和を求める関数 mySum を定義せよ。

問 1.6.2 真偽値のリスト [Bool] を2進数と見なして、対応する整数を計算する関数 fromBin :: [Bool] -> Integer を定義せよ。例えば、fromBin [True, True] は3, fromBin [True, False, True, False] は10になる。

ヒント: 引数の数を一つ増やした補助関数が必要になる。

問 1.6.3 実数のリスト xs と実数から実数への関数 f を受け取り、リストの各要素に f を適用した結果の和を計算する関数 sumf :: [Double] -> (Double -> Double) -> Double を定義せよ。

## 1.7 帰納法による証明

プログラムを関数型言語で表現することの利点は、プログラムの同値性(等価性)などの議論が容易になるというところにある。ここでは、そのような議論の例として、2つのリストに関する関数が等価であることの証明を取り上げる。このような証明は帰納法と併用することが多い。以下の例も帰納法を使用している。

リストを反転する関数 reverse:

```
append      :: [a] -> [a] -> [a];  
append [] ys      = ys;  
append (x:xs) ys = x : (append xs ys);  
  
reverse     :: [a] -> [a];  
reverse []      = [];  
reverse (x:xs) = append (reverse xs) [x];
```

は上の定義では、引数の \_\_\_\_\_ に比例する時間がかかるために効率が悪い。

そこで次のような定義を考える。

```
-- shunt は rev の補助関数
shunt      :: [a] -> [a] -> [a];
shunt ys []      = ys;
shunt ys (x:xs) = shunt (x:ys) xs;

rev      :: [a] -> [a];
rev xs = shunt [] xs;
```

この rev という関数は、引数の \_\_\_\_\_ に比例する時間でリストを反転できるので効率が良い。

rev と reverse が等価であること – 正確に言うと、すべての有限リスト xs に対して

```
rev xs = reverse xs
```

が成り立つことを証明できる。

そのためには、次のような補助定理を証明すれば良い。

```
shunt ys xs = append (reverse xs) ys
```

これは、 \_\_\_\_\_ で証明することができる。

証明:

xs = [] のとき:

---

---

---

---

---

xs = z:zs のとき:

---

---

---

---

---

問 1.7.1 すべての有限リスト xs について、

```
append xs (append ys zs) = append (append xs ys) zs
```

が成り立つことを、xs に関する帰納法で証明せよ。

---

---

---

---

---

## 1.8 有用なリストの高階関数

次のようなリストに対する高階関数がよく利用される。これらは Prelude ( 標準ライブラリ ) に定義済みである。

```
map :: (a -> b) -> [a] -> [b]
map f []      = []
map f (x:xs) = f x : map f xs

filter :: (a -> Bool) -> [a] -> [a]
filter p []    = []
filter p (x:xs) = if p x then x : filter p xs else filter p xs

iterate :: (a -> a) -> a -> [a]
iterate f x = x : iterate f (f x)

foldr :: (a -> b -> b) -> b -> [a] -> b
foldr f x []    = x
foldr f x (y:ys) = f y (foldr f x ys)

foldl :: (a -> b -> a) -> a -> [b] -> a
foldl f x []    = x
foldl f x (y:ys) = foldl (f x y) ys
```

## 1.9 その他の便利な記法

関数の中置記法化 Haskell の関数は通常は前置記法で用いるが、識別子を “\_” ( バッククオート、クオート ( “” ) ではないことに注意 ) で囲むことによって、中置記法で書くことができる。これは小さなことに見えるが実は意外に便利である。例えば、次のように定義された zip の場合、

```
zip :: [a] -> [b] -> [(a,b)];
zip (a:as) (b:bs) = (a,b) : zip as bs;
zip _ _ _ = [];
```

通常は `zip [1, 2] [3, 4]` のように書くが、これを `[1, 2] `zip` [3, 4]` と書いても良い。

中置記法で用いる演算子に対して、`infixl`, `infixr`, `infix` などのキーワードを使って、優先順位と結合性を定めることができる。たとえば、Prelude ( Haskell はじめから読み込まれる標準ライブラリ ) では次のように宣言されている。

```
infixr 9 .;
```

```
infixl 9  !!;
infixr 8  ^, ^~, **;
infixl 7  *, /, `quot`, `rem`, `div`, `mod`, :%, %;
infixl 6  +, -;
infixr 5  ::;
infixr 5  ++;
infix  4  ==, /=, <, <=, >=, >, `elem`, `notElem`;
infixr 3  &&;
infixr 2  ||;
infixl 1  >>, >>=;
infixr 1  =<<;
infixr 0  $, $!, `seq`;
```

infixl は \_\_\_\_\_、 infixr は \_\_\_\_\_ を表す。ただの infix はどちらでもないこと（非結合—例えば  $1 < x < 2$  は構文エラー！）を表す。また、2列目の数字が大きいほど、優先順位が高い。例えば\*は + よりも結合力が強い。

バッククオートと逆に本来中置記法で使用される演算子を“(”と“)”でくくって、ふつうの前置記法で用いることができる。例えば $1+2$ を $(+) 1 2$ と書くことができる。あるいは関数の引数として渡すこともできる。

局所的定義 `let` というキーワードを用いて、局所的な変数を定義することができる。

**let** (複数の) 变数の定義 **in** 式

という形で用いる。

```
pow4 x  = let { y = x*x } in y*y;
head ys = let { (x:xs) = ys } in x;
```

などである。この例では4乗する関数（pow4）を定義しているが、 $y^*y$  の部分が変数 y の有効範囲（\_\_\_\_\_）に属している。実は、さらに“変数の定義”の右辺の部分（この例では、\_\_\_\_\_の部分）もスコープに属している。これは次の例でわかる。

```
repeat  :: a -> [a];
repeat x = let { xs = x:xs } in xs;
```

この関数は要素  $x$  の無限リストを生成する。

(このような定義は、Haskellでは“止まらない・役に立たない式”ではなく、意味のある式となる。このことは、あとで Haskell の評価戦略を紹介する時に説明する。)

問 1.9.1 リストを昇べきの順に表された多項式と見なし、多項式の値を計算する関数 `evalPoly :: [Double] -> Double -> Double` を定義せよ。

問 1.9.2 リストを集合だと見なして、そのベキ集合（部分集合の集合）を返す関数 powerset を定義せよ。

例えば powerset `[1, 2, 3]` は `[[], [1], [2], [3], [1, 2], [2, 3], [1, 3], [1, 2, 3]]` になる。(ただし、順番はこの通りでなくても良い。)

<sup>4</sup>Ctrl-c で中断する。

## 1.10 代数的データ型の定義

リストのようなデータ型をプログラマがあらたに定義する方法も用意されている。このようなデータ型は複数の \_\_\_\_\_ を持つことができる。このようなデータ型を \_\_\_\_\_ ( algebraic datatype ) という。

データ型の宣言の一般的な形式は次のとおりである。

```
data 型構成子名 型引数1 型引数2 ... 型引数k
  = 構成子名1 型1,1 ... 型1,n1
  | 構成子名2 型2,1 ... 型2,n2
  |
  | ...
  | 構成子名m 型m,1 ... 型m,nm
;
```

型構成子名・構成子名ともに使える文字は変数名の場合と同じだが、変数名とは逆に \_\_\_\_\_ から始まる必要がある。

代数的データ型はパラメータがない場合は、C の列挙 ( enum ) 型と同じようなものである。次の例では、

```
data Direction = Up | Down | Left | Right
```

Up, Down, Left, Right の 4 つが Direction 型を構成している。

一般的には代数的データ型の構成子はパラメータを持つ。次の例は二分木を表すデータ型を定義している。

```
data Tree a = Branch (Tree a) a (Tree a) | Empty
```

このデータ型は Branch と Empty の 2 つの構成子を持つ。Empty は引数を取らず、それだけで二分木を構成する。Branch は 3 つのパラメータを取る。1 番目と 3 番目は自分自身と同じ型の二分木であり、2 番目は要素である。つまり、Branch は次のような型を持っている。


Branch :: \_\_\_\_\_

ここで、a は \_\_\_\_\_ であり、ここには Integer や String などの具体化な型がはいることができる。例えば Tree Integer は要素が Integer 型であるような二分木の型である。Tree 自体は型ではなく型構成子 (type constructor) である。つまり、型パラメータを伴ってはじめて型になる。

具体的には次のような式がこの Tree 型を持つ。

```
Empty :: Tree a
Branch Empty 1 Empty :: Tree Integer
Branch (Branch Empty "a" Empty) "b" (Branch Empty "c" Empty) :: Tree String
Branch (Branch Empty 1 Empty) 2 (Branch Empty 3 (Branch Empty 4 Empty))
:: Tree Integer
```

例えば最後の式の構造は、図で表すと次のようになる。



問 1.10.1 Tree 型に対して、次のような関数を定義せよ。

```

size      :: Tree a -> Integer -- 要素数
depth     :: Tree a -> Integer -- 深さ
preorder  :: Tree a -> [a]     -- 前順走査
postorder :: Tree a -> [a]     -- 後順走査
reflect   :: Tree a -> Tree a -- 鏡像

```

## 1.11 Haskell の評価戦略

Haskell の評価戦略は、必要になるまで引数の評価を遅らせる \_\_\_\_\_ ( lazy evaluation ) と呼ばれる戦略である。 \_\_\_\_\_ と呼ばれる手法と組み合わせて、同じ計算の繰り返しを避けるようになる。

例えば、次のように定義された `double` という関数を考える。


```
double x = x+x
```

グラフ簡約を用いない遅延評価 (?) では `double (2+2)` という式は次のように計算することになる。

$$\begin{aligned}
 \text{double} (2+2) &= (2+2) + (2+2) \\
 &= 4 + 4 \\
 &= 8
 \end{aligned}$$

つまり、`double` の引数である `(2+2)` は計算されないまま、まず `double` の定義にしたがって式が展開される。そして、本当に必要になって（この場合は別の+の引数だから必要）はじめて `2+2` が計算される。この方式をナイーブに実行すると、`2+2` が二度計算されてしまう。

グラフ簡約では、このような計算をグラフの形で表して、`2+2` を一度しか計算しないようにしている。



遅延評価の良いところは概念的に無限の大きさのデータ構造を扱えることである。例えば次のような関数を考える。

```

from :: Integer -> [Integer];
from n = n : from (n+1);

take :: Integer -> [a] -> [a];
take 0 _      = [];
take _ []     = [];
take n (x:xs) = x : take (n-1) xs;

```

すると `from 1` は `[1, 2, 3, ... ]` という無限リストだが、この無限リストを途中で用いている `take 3 (from 1)` という式は

```
take 3 (from 1) → take 3 (1:from (1+1)) → 1:(take 2 (from (1+1)))
→ 1:(take 2 ((1+1):from (1+1+1))) → 1:(1+1):(take 1 (from (1+1+1))) → ...
→ 1:(1+1):(1+1+1):(take 0 (from (1+1+1+1))) → 1:(1+1):(1+1+1):[] (=[1, 2, 3])
```

のように有限時間で計算できる。

なお、`from` で生成されるような等差数列については.. を使った略記法がいくつか用意されている。

```
Prelude> [1..]
[1,2,3,4,5,6,7,8,9,10,11, ...
Prelude> [2,4..]
[2,4,6,8,10,12,14,16,18,20,22, ...
Prelude> [1..10]
[1,2,3,4,5,6,7,8,9,10]
Prelude> [1,4..20]
[1,4,7,10,13,16,19]
```

遅延評価を用いるといろいろと興味深いプログラミングが可能になる。参考文献 [6] は、遅延評価を利用したプログラムの部品化の手法を詳しく説明している。

問 1.11.1 `take` の反対に、リストの最初の  $n$  個の要素を取り除く関数 `myDrop` を定義せよ。

問 1.11.2 `fib` をフィボナッチ数列の無限リストとして定義せよ。

問 1.11.3 要素に重複のない昇順に並んだ 2 つのリストをマージして、やはり重複なしの昇順のリストを生成する関数 `merge` を定義せよ。

問 1.11.4  $2^i \cdot 3^j \cdot 5^k$  ( $i, j, k$  は 0 以上の整数) の形で表せる整数のみを重複なしに昇順に並べたリスト `hamming` を定義せよ。

問 1.11.5 Haskell 以外の言語で、無限リストをシミュレートする方法を考察せよ。またそれを用いて、素数列を生成せよ。

## 1.12 リストの内包表記 (List Comprehension)

Haskell は、リストの \_\_\_\_\_ (List Comprehension) という糖衣構文 (syntax sugar) を持つ。これは数学で使われる集合の内包表記に似た記法である。

例

```
Prelude> [(x, y) | x <- [1, 2, 3, 4], y <- [5, 6, 7]]
[(1,5),(1,6),(1,7),(2,5),(2,6),(2,7),(3,5),(3,6),(3,7),(4,5),(4,6),(4,7)]
Prelude> [x*x | x <- [1..10], odd x]
[1,9,25,49,81]
```

(ただし `[1..10]` は `[1,2,3,4,5,6,7,8,9,10]` の略記法である。)

リストの内包記法は次のような形のものである。

[ 式 | 限定式, ... , 限定式 ]

ここで限定式は、Bool 型の式 (ガード) か、次の形 (生成式) :

変数（一般にはパターン）  $\leftarrow$  式

のいずれかである。生成式の左辺の変数のスコープは、それより後の限定式である。生成式で変数に右辺の式を評価して得られるリストの要素を順に代入し、ガードで真となるもののみ抽出して、すべての組み合わせを列挙する。

問 1.12.1 非負の整数  $n$  を受け取り、 $0 \leq x \leq y \leq n$  となるすべての  $x, y$  の組を生成する関数 `foo :: Integer -> [(Integer, Integer)]` を定義せよ。

問 1.12.2 非負の整数  $n$  を受け取り、 $0 < x < y < z \leq n$  の範囲で  $x^2 + y^2 = z^2$  となるすべての  $x, y, z$  の組を生成する関数 `chokkaku :: Integer -> [(Integer, Integer, Integer)]` を定義せよ。

翻訳 リストの内包表記は次のような高階関数を用いて翻訳することができる。

```
-- 要素数 1 のリストを返す
unit :: a -> [a];
unit a = a : [];

bind :: [a] -> (a -> [b]) -> [b];
bind [] _ = [];
bind (x:xs) f = append (f x) (bind xs f);
```

翻訳規則は次のとおりである。

```
[t]    => unit t
[t | x <- u, P] => bind u (\ x -> [t | P])
[t | b, P]    => if b then [t | P] else []
```

ただし、 $b$  は Bool 値の式、 $P$  は限定式のならびである。

内包表記を用いると、クイックソートは次のように簡潔に表される。

```
qsort []      = []
qsort (x:xs) = qsort [ y | y <- xs, y < x] ++ x : qsort [ y | y <- xs, y >= x]
```

問 1.12.3 次の内包記法を上の翻訳規則を用いて、`unit, bind` を用いた形にせよ。

1.  $[(x, y) | x <- [1, 2, 3, 4], y <- [5, 6, 7]]$
2.  $[x*x | x <- [1..10], \text{odd } x]$

問 1.12.4 `primes` を素数列  $[2, 3, 5, 7, 11, \dots]$  の無限リストとして定義せよ。

考え方：“エラトステネス (Eratosthenes) のふるい” というアルゴリズムを実装する。このおなじみのアルゴリズムを言葉で表現すると次のようになる。

1. 2 以上の自然数を並べる。
2. 先頭の数を取り除き、その倍数を同時にとり除く。
3. 2 を繰り返す。

この時に先頭に現れた数を順番に並べたものが素数の列である。途中で無限リストの無限リストが現われるが、問題ない。


このようにして、素数を無限リストとして表現することで、さまざまな“境界条件”に対応することができる。Cなどで実装しようとすると、1000までの素数というのは配列を用いて簡単に求めることができるが、最初の100個の素数を求めるのは急に難しくなる。

### 1.13 8クイーンの問題

リストの内包表記を用いて、有名なパズルを解いてみることにする。

8クイーンの問題は8個のクイーンを、お互いにとることができないように、チェス盤の上に置くという問題である。可能な解はいくつか存在する。この可能な解の集まりをリストとして表現する。ここでは無限リストは本質的には使用しないが、遅延評価は効率の点で決定的な役割を果たす。

クイーンの配置は、ここでは数のリストで表す。[4, 6, 1, 5, 2, 8, 3, 7] は次のような配置を表す。



safe p n は、length p 列までのクイーンの配置が p というリストで与えられた時、第 length p + 1 列の第 n 行にクイーンを置くことができるかどうかを示す関数である。

```
safe p n = all not [ check (i, j) (1+length p, n) | (i, j) <- zip [1..] p ]
check (i, j) (m, n) = j==n || (i+j==m+n) || (i-j==m-n)
```


ここで、all は

```
all          :: (a -> Bool) -> [a] -> Bool
all p []    = True
all p (x:xs) = if p x then all p xs else False
```

と定義された標準ライブラリの関数である。[1..] は [1, 2, 3, ...] という無限リストである。

```
> safe [1, 3] 5
True
> safe [1, 3] 2
False
```

となる。



順に、最初の m 列のすべての安全な配置を調べていく、そのために、そのようなすべての配置(のリスト)を返す queens という関数を定義する。

```
queens 0 = []
queens m = [ append p [n] | p <- queens (m-1), n <- [1..8], safe p n ]
```

例えば

```
> queens 1
[[1], [2], [3], [4], [5], [6], [7], [8]]
> queens 2
[[1,3],[1,4],[1,5],[1,6],[1,7],[1,8],[2,4],[2,5],[2,6],[2,7],[2,8],
 [3,1],[3,5],[3,6],[3,7],[3,8],[4,1],[4,2],[4,6],[4,7],[4,8],[5,1],
 [5,2],[5,3],[5,7],[5,8],[6,1],[6,2],[6,3],[6,4],[6,8],[7,1],[7,2],
 [7,3],[7,4],[7,5],[8,1],[8,2],[8,3],[8,4],[8,5],[8,6]]
```

となる。そうすると `head (queens 8)` で最初の解を求めることができる。

```
> head (queens 8)
[1,5,8,6,3,7,2,4]
```

遅延評価を用いているので、最初の解を求めるためには本当に必要な部分の簡約しか行なわない。つまり、上の `[1, 5, 8, 6, 3, 7, 2, 4]` を求めるのに、`queens 7` の計算をすべて行なっているわけではなく、この最初の解を求めるのに必要なだけの部分の計算をしている。これは、`queens 7` を完全に計算させると `head (queens 8)` よりも計算に時間がかかることからもわかる。ここでは、遅延評価は Prolog いうところの後戻り (backtrack) と同じような効果を実現する。1 つだけ解を求める計算とすべての解を求める計算を別々に記述する必要はなく、しかも、1 つだけ解を求めるためには、それに必要なだけの計算しか行なわない

ちなみに `queens 8` を計算させると、

```
> queens 8
[[1,5,8,6,3,7,2,4],[1,6,8,3,7,4,2,5],
(略)
,[8,3,1,6,2,5,7,4],[8,4,1,3,6,2,7,5]]
```

解はすべてで 92 個あることがわかる。

## この章の参考文献

- [1] 「Haskell – A Purely Functional Language featuring static typing, higher-order functions, polymorphism, type classes and monadic effects」  
<http://www.haskell.org/>  
Haskell に関するもっとも主要な情報源である。
- [2] 「Programming in Haskell」<http://www.sampou.org/cgi-bin/haskell.cgi>  
日本語での Haskell の主要な情報源である。
- [3] Simon Peyton Jones, John Hughes 他 「Haskell 98: A Non-strict, Purely Functional Language」  
1999 年 2 月, <http://www.haskell.org/onlinereport/>  
Haskell の仕様書、Haskell を利用する上での基本ドキュメントである。
- [4] Mark P Jones, Alastair Reid 他 「The Hugs 98 User Manual」  
<http://cvs.haskell.org/Hugs/pages/hugsman/>  
もうひとつの Haskell のメジャーな処理系 Hugs のユーザーマニュアルである。
- [5] Simon Peyton Jones, David Lester 「Implementing Functional Languages」  
Prentice Hall, 1992 年  
<http://research.microsoft.com/Users/simonpj/Papers/pj-lester-book/>  
Haskell の実装について知りたい人にお勧めする。
- [6] John Hughes 「Why Functional Programming Matters」  
1989 年, <http://www.md.chalmers.se/~rjmh/Papers/whyfp.html>  
遅延評価を実際のプログラムでどのように用いるかを解説している。

日本語訳 – 山下 伸夫 訳 「なぜ関数プログラミングは重要か」  
<http://www.sampou.org/haskell/article/whyfp.html>

- [7] Philip Wadler 「List Comprehensions」  
( Simon Peyton Jones 「The Implementation of Functional Programming Languages」  
Prentice Hall, 1987 年  
<http://research.microsoft.com/users/simonpj/Papers/slpj-book-1987/>  
のなかの第 7 章 )  
リストの内包表記を解説している。
- [8] 和田 英一 他 「Haskell プログラミング」  
<http://www.ipsj.or.jp/07editj/promenade/>  
情報処理学会の会誌「情報処理」に 2005 年 4 月から 2006 年 3 月まで連載された。
- [9] Simon Peyton Jones 「Wearing the hair shirt – A retrospective on Haskell」  
2003 年, <http://research.microsoft.com/~simonpj/papers/haskell%2Dretrospective/>  
Haskell を設計した中心人物の一人による Haskell の設計の “回顧” ( と “展望” ) である。
- [10] Richard Bird, Philip Wadler 著 武市 正人 訳 「関数プログラミング」  
近代科学社, 1991 年 4 月, ISBN4-7649-0181-1  
Haskell でなく、Miranda を使っているがとても有名な良書である。
- [11] 向井 淳 「入門 Haskell はじめて学ぶ関数型言語」  
毎日コミュニケーションズ, 2006 年 3 月, ISBN4-8399-1962-3
- [12] 山下 伸夫 (監) 青木 峰郎 (著)  
「ふつうの Haskell プログラミング ふつうのプログラマのための関数型言語入門」  
ソフトバンク クリエイティブ, 2006 年 6 月, ISBN4-7973-3602-1